Лабораторная работа 1

Общий подход к созданию приложений,

 связанных с вычислениями.

Цель настоящей работы – показать общий подход к созданию приложений, связанных с вычислениями на примере разработки приложения «Решение квадратного уравнения». Сама по себе задача решения квадратного уравнения тривиальна и, мы надеемся, что Вы уже сможете сделать это самостоятельно.

 Однако, если Вам необходимо разработать действительно удобное для работы приложение, которое будет не только что-то вычислять, но также проверять наличие необходимых данных для расчета и осуществлять контроль вводимой информации, например, не допускать ввод неверных (нечисловых) данных, то задача уже не выглядит столь примитивной. Общий подход, приведенный в настоящей лабораторной работе может пригодиться для создания любого серьезного приложения.

Задание 1. Сначала попробуйте самостоятельно создать приложение под названием «Решение квадратного уравнения», в котором были бы три окна для ввода коэффициентов уравнения, три кнопки (<Вычислить>, <Новое>, <Выход>) и поле для вывода результата согласно рис. 1.1. В качестве результата должен расчитываться дискриминант и выводиться сообщение о наличии корней уравнения и их значения. 

Сохраните проект в отдельной папке и покажите результат работы преподавателю.

Рис.  1.1. Вид диалогового окна приложения

                «Решение квадратного уравнения»

Примечание. Поле метки Label (прямоугольник под названием «Дискриминант») предназначено для вывода результата.

Задание 2. Усовершенствование программы.

Постановка задачи.

1. В поля коэффициентов уравнения допускается ввод только целых и дробных положительных и отрицательных чисел. Целая часть числа отделяется от дробной запятой. Ввод других символов программа должна блокировать.

2. При нажатии клавиши <Enter> после ввода очередного коэффициента уравнения, курсор должен автоматически перемещаться в поле ввода следующего коэффициента. Если клавиша <Enter> нажата после ввода последнего коэффициента, программа должна автоматически перейти к вычислениям.

3. Должен осуществляться контроль за наличием всех вводимых данных. Если данных не хватает, об этом выводится сообщение и программа блокируется.

4. Должна осуществляться проверка, является ли уравнение квадратным. Если не является – об этом выдается сообщение и программа блокируется.

5. Программа должна иметь три управляющих кнопки: <Вычислить>, <Новое>, <Выход>.

6. В качестве результата должен рассчитываться дискриминант и выводиться сообщение о наличии корней и их значения.

Выполните следующие действия:

· откройте новый проект;

· форме дайте название «Решение квадратного уравнения»;

· для ввода коэффициентов уравнения поместите на форму компоненты Edit1, Edit2 u Edit3;
· для управления приложением поместите на форму компоненты  Button1, Button2, Button3 и закончите оформление формы согласно рис. 1.1;

· разберитесь и проанализируйте процедуру обработки события OnKeyPress (Событие по нажатию клавиши) для поля редактирования Edit1, приведенную ниже, которая в зависимости от того, какую клавишу нажал пользователь, будет выполнять следующие  действия:

· если нажата цифровая клавиша, то в результате выполнения процедуры в поле редактирования появляется символ 
     цифры;
· если нажата клавиша <BackSpace>, то последний введённый символ стирается;
· если нажата «запятая», то процедура проверяет, есть ли уже в поле редактирования такой символ. Если есть, то появление второй «запятой» блокируется;
· если нажата клавиша <->, то минус появляется только в том случае, если в поле редактирования нет ни одной цифры;
· если нажата клавиша <Enter>, то применением метода SetFocus к очередному полю редактирования Edit курсор перемещается в это поле или на кнопку <Вычислить>;
· остальные клавиши блокируют ввод, передавая параметру Key процедуры значение chr(0).
В этой процедуре, описанной ниже, приведен классический пример использования оператора варианта. Примеры использования условных операторов неоднократно встречаются во всей программе и в комментариях не нуждаются.
procedure TForm1.Edit1KeyPress(Sender:TObject; var Key: Char);
// Заголовок процедуры показывает, что процедура в качестве 

// выходного параметра получает символьную переменную Key,

// которая содержит код нажатой  клавиши.

var BUF:string[20];                       //Переменная-буфер

begin

{--------------------- Выбор окна ввода для дальнейшей обработки -------------}

   if Sender=Edit1 then BUF:=Edit1.Text;

   if Sender=Edit2 then BUF:=Edit2.Text;

   if Sender=Edit3 then BUF:=Edit3.Text;

{---------------------------- Обработка строки ввода (буфера) ---------------------}

  case Key of
   '0'..'9',chr(8):;          //Для этих символов ввод разрешен

   ','            :if pos(',',BUF)<>0 then Key:=chr(0);
  '-'           :if length(BUF)<>0  then Key:=chr(0);

                                          //Блокировка ввода

   chr(13)        :begin
 {---------------------------------- Установка фокуса ----------------------------------}

                                        if Sender=Edit1 then Edit2.SetFocus;

                    if Sender=Edit2 then Edit3.SetFocus;

                    if Sender=Edit3 then Button1.SetFocus;

                 end;

 '.'            :begin
{--------------------------------- Выдача сообщения -----------------------------------}

                    MessageDlg('Для отделения целой части числа'

                    +#13+'от дробной используйте запятую',

                    mtInformation,[mbOk],0);

                    Key:=chr(0)           //Блокировка ввода

                   end

    else Key:=chr(0)          //Блокировка ввода для любых других символов

  end;

end;
Примечания.

1. Параметр Sender является указателем на объект программы, активный в данный момент времени. Таким образом, информацию об источнике событий можно получить проверкой значения параметра Sender.

2. Переменная Key в заголовке процедуры обработки события OnKeyPress, является её параметром и содержит код клавиш, нажатой на клавиатуре. Поэтому, присвоив Key=chr(0) можно блокировать ввод или заменить значение на другой символ, передавая значению Key код символа замены.

3. Переменная BUF служит буфером, в котором накапливаются данные из строк ввода и, далее, анализируются.

· процедуру обработки события OnKeyPress для поля редактирования Edit1 наберите согласно коду, приведенному выше (комментарии можно опустить);
· очевидно, что процедуры обработки события OnKeyPress для разных полей редактирования отличаются только их реакцией на нажатие клавиши <Enter>, а в остальном - они идентичны. Поэтому при создании процедур обработки события OnKeyPress для следующих полей редактирования Edit2 и Edit3, выберете уже созданную процедуру для Edit1 из раскрывающегося списка (событие OnKeyPress на закладке Events Инспектора Объектов). Таким образом, эти процедуры не нужно создавать заново для каждого объекта, а можно просто назначить их на основе уже созданной процедуры;
· теперь Вам осталось написать обработку нажатий кнопок. Ниже приведён текст модуля полностью с необходимыми комментариями. 

Примечание. Подчёркнутым шрифтом выделено то, что предстоит написать Вам, обычным шрифтом – заготовка, созданная Delphi. Служебные слова выделены жирным шрифтом.
unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls,

  Forms, Dialogs, StdCtrls;

type

  TForm1 = class(TForm)

    Label1: TLabel;

    Label2: TLabel;

    Label3: TLabel;

    Label4: TLabel;

    Edit1: TEdit;

    Edit2: TEdit;

    Edit3: TEdit;

    Label5: TLabel;

    Label6: TLabel;

    Button1: TButton;

    Button2: TButton;

    Button3: TButton;

    procedure Edit1KeyPress(Sender: TObject; var Key: Char);

    procedure Button1Click(Sender: TObject);

    procedure Button2Click(Sender: TObject);

    procedure Button3Click(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation

{$R *.DFM}

{=============== Обработка события <OnKeyPress> =============}

procedure TForm1.Edit1KeyPress(Sender:TObject; var Key: Char);

var BUF:string[20];                       //Переменная-буфер

begin

{------------------- Выбор окна ввода для дальнейшей обработки ---------------}

   if Sender=Edit1 then BUF:=Edit1.Text;

   if Sender=Edit2 then BUF:=Edit2.Text;

   if Sender=Edit3 then BUF:=Edit3.Text;

{-------------------------- Обработка строки ввода (буфера) -----------------------}

  case Key of
   '0'..'9',chr(8):;          //Для этих символов ввод разрешен

   ','            :if pos(',',BUF)<>0 then Key:=chr(0);

  '-'            :if length(BUF)<>0  then Key:=chr(0);

                                          //Блокировка ввода

  chr(13)        :begin
 {------------------------------- Установка фокуса -------------------------------------}

                    if Sender=Edit1 then Edit2.SetFocus;

                    if Sender=Edit2 then Edit3.SetFocus;

                    if Sender=Edit3 then Button1.SetFocus;

                   end;

  '.'            :begin

   {----------------------------- Выдача сообщения -------------------------------------}

                    MessageDlg('Для отделения целой части числа'

                    +#13+'от дробной используйте запятую',

                    mtInformation,[mbOk],0);

                    Key:=chr(0)           //Блокировка ввода

                   end

    else Key:=chr(0)  //Блокировка ввода для любых других символов

  end;

end;
{================== Кнопка <Новое> ========================}

procedure TForm1.Button2Click(Sender: TObject);

begin

   Edit1.Text:='';                       //Очистка строк ввода

     Edit2.Text:='';

   Edit3.Text:='';

   Label5.Caption:='';                //Очистка поля дискриминанта

   Label6.Font.Color:=clBlack;    //Цвет шрифта сообщения

   Label6.Caption:='';                   //Очистка поля решения

   Edit1.SetFocus;                           //Установка фокуса

end;

{================== Кнопка <Вычислить> ====================}

procedure TForm1.Button1Click(Sender: TObject);

var A,B,C:real;                        //Коэфициенты уравнения

    D:real;                                    //Дискриминант

    X1,X2:real;                          //Корни уравнения

begin

{---------------------------- Проверка наличия данных -------------------------------}

 if (Edit1.Text='')or(Edit2.Text='')or(Edit3.Text='') then
     begin

       MessageDlg('Нужно задать все'+#13

                +'коэффициенты уравнения',mtWarning,[mbOk],0);

       Edit1.SetFocus;                  //Установка фокуса
     end

     else

{------------------------------------- Р А С Ч Е Т ---------------------------------------}

     begin

       A:=StrToFloat(Edit1.Text);    //Ввод коэффициентов уравнения

       B:=StrToFloat(Edit2.Text);

       C:=StrToFloat(Edit3.Text);
{------------------- Проверка: является ли уравнение квадратным ---------------}

       if A=0 then
         begin

           Label6.Font.Color:=clRed;   //Цвет шрифта сообщения

                      Label6.Caption:=

           'Коэффициенент при второй степени равен нулю!';

           Edit1.SetFocus;              //Установка фокуса

                  end

         else

         begin

         D:=Sqr(B)-4*A*C;            //Вычисление дискриминанта

                  Label5.Caption:='ДИСКРИМИНАНТ='

                         +FloatToStrF(D,ffGeneral,7,4);

         if D<0 then
           begin

             Label6.Font.Color:=clRed;  //Цвет шрифта сообщения

                          Label6.Caption:='Уравнение корней не имеет';

           end

           else

           begin
             X1:=(-B+Sqrt(D))/(2*A);    //Вычисление корней

                          X2:=(-B-Sqrt(D))/(2*A);

             Label6.Font.Color:=clBlue; //Цвет шрифта сообщения

                           Label6.Caption:='Корни уравнения:'

               +#13+'X1='+FloatToStrF(X1,ffGeneral,7,4)

               +#13+'X2='+FloatToStrF(X2,ffGeneral,7,4);

           end;

         end;

     end;

end;

{==================== Кнопка <ВЫХОД> ====================}

procedure TForm1.Button3Click(Sender: TObject);

begin

 Close;

end;

end.

· сохраните результаты Вашей работы в отдельном каталоге. На запрос Delphi о задании имен файлов введите для проекта – Kvur.dpr, а имя файла программного кода (Unit1.pas) оставьте без изменения;

· запустите проект. Попробуйте различные варианты ввода исходных данных (в том числе и ошибочные). Проверьте, как на них реагирует программа.

Работа с различными форматами изображения чисел

Для вывода вычисленных значений корней уравнения в программе используется функция FloatToStrF, которая преобразует дробное число в его строковое представление и позволяет получать изображение числа в различных форматах. В общем виде обращение к функции FloatToStrF выглядит так:

S:= FloatToStrF (<Значение>, <Формат>, <Точность>, <КоличествоЦифр>),

где:

       S – строковое представление (изображение) числа <Значение>;

       <Значение> – число, изображение которого надо получить;

       <Формат> – именованная константа, определяющая вид строки S, 

                            согласно таблице, приведенной ниже:

Константа
Формат отображения числа

ffGeneral
Общий цифровой формат.

ffExponent
Научный формат.

ffFixed
Формат с десятичной точкой.

ffNumber
Числовой формат.

ffCurrency
Денежный формат.

       <Точность> – параметр, определяющий точность представления 

                            преобразуемого числа. Значение <Точность> должно быть:

                                    ( 7 для типа Single;

                                    ( 15 для типа Double;

                                    ( 18 для типа Extended;

       <КоличествоЦифр> – значение этого параметра зависит от используе-

                                мого формата.

Задание 3. Выполнить следующее:

· поэкспериментируйте с различными форматами изображения чисел;

· результаты всей проделанной Вами работы сохраните в отдельной папке, задав проекту имя KVUR.dpr и продемонстрируйте преподавателю.
Вопросы

1. Каков общий подход к контролю за вводом исходных данных в приложениях, предназначенных для вычислений?

2. Каким путём процедура обработки события может заменить введённый символ или блокировать его появление в поле редактирования?

3. Поясните принцип работы обработчика событий OnKeyPress в программе этой лабораторной работы. Каково назначение переменной-буфера в этой процедуре?

4. Каково назначение параметра Sender в процедурах Delphi?
5. Каково назначение метода SetFocus?
6. Каким образом можно организовать одинаковые процедуры обработки события OnKeyPress для разных полей редактирования, записав программный код только один раз? Можно ли выполнить такую операцию для других объектов программы? Других процедур?

7. Какая функция используется для преобразования дробного числа в его строковое представление и позволяет получать изображения числа в разных форматах? Опишите общий вид обращения к этой функции.

8. Какие форматы изображения числа Вы знаете?

Резюме

1. Работа условного оператора и оператора вариант в Delphi осуществляются так же, как в Турбо Паскале.

2. Показан общий подход к контролю за вводом исходных данных в приложениях, предназначенных для вычислений, состоящий в проверки их наличия и блокирующий неправильный набор. Этот подход реализован в обработчике событий OnKeyPress программы настоящей лабораторной работы.

3. Объект Sender, являющийся параметром любой процедуры Delphi, является указателем на объект программы, активный в конкретный момент времени. Таким образом, информацию об источнике событий можно получить проверкой значения параметра Sender.

4. Метод SetFocus перемещает курсор на объект, к которому он применён, и делает его текущим на форме.

5. Программный код процедуры обработки события можно записать для какого-либо объекта программы только один раз, а затем назначить его для другого объекта, выбрав это событие (если оно для него предусмотрено) из раскрывающегося списка на странице Events Инспектора Объектов.

6. Для вывода дробного числа корней уравнения используется функция FloatToStrF, которая преобразует это число в его строковое представление и позволяет получать изображение числа в различных форматах.

